

F03ADF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F03ADF calculates the determinant of a complex matrix using an LU factorization with partial pivoting.

2 Specification

```
SUBROUTINE F03ADF(A, IA, N, DETR, DETI, WKSPCE, IFAIL)
INTEGER          IA, N, IFAIL
  real          DETR, DETI, WKSPCE(*)
  complex      A(IA,*)
```

3 Description

This routine calculates the determinant of A using the LU factorization with partial pivoting, $PA = LU$, where P is a permutation matrix, L is lower triangular and U is unit upper triangular. The determinant of A is the product of the diagonal elements of L , with the correct sign determined by the row interchanges.

4 References

- [1] Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer-Verlag

5 Parameters

- 1:** A(IA,*) — *complex* array *Input/Output*
Note: the second dimension of the array A must be at least $\max(1,N)$.
On entry: the n by n matrix A .
On exit: A is overwritten by the factors L and U , except that the unit diagonal elements of U are not stored.
- 2:** IA — INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F03ADF is called.
Constraint: $IA \geq \max(1,N)$.
- 3:** N — INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.
- 4:** DETR — *real* *Output*
- 5:** DETI — *real* *Output*
On exit: the real part and imaginary parts of the determinant of A .
- 6:** WKSPCE(*) — *real* array *Workspace*
Note: the dimension of the array WKSPCE must be at least $\max(1,N)$.

7: IFAIL — INTEGER*Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors detected by the routine:

IFAIL = 1

The matrix A is singular, possibly due to rounding errors. The factorization could not be completed. DETR and DETI are set to 0.0.

IFAIL = 2

Overflow. The value of the determinant is too large to be held in the computer.

IFAIL = 3

Underflow. The value of the determinant is too small to be held in the computer.

IFAIL = 4

On entry, $N < 0$,
or $IA < \max(1,N)$.

7 Accuracy

The accuracy of the determinant depends on the conditioning of the original matrix. For a detailed error analysis, see Wilkinson and Reinsch [1] page 107.

8 Further Comments

The time taken by the routine is approximately proportional to n^3 .

9 Example

To calculate the determinant of the complex matrix:

$$\begin{pmatrix} 1 & 1 + 2i & 2 + 10i \\ 1 + i & 3i & -5 + 14i \\ 1 + i & 5i & -8 + 20i \end{pmatrix}$$

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F03ADF Example Program Text
*      Mark 15 Revised.  NAG Copyright 1991.
*      .. Parameters ..
      INTEGER          NMAX, IA
      PARAMETER       (NMAX=8,IA=NMAX)
      INTEGER          NIN, NOUT
      PARAMETER       (NIN=5,NOUT=6)
```

```

*    .. Local Scalars ..
  real          DETI, DETR
  INTEGER       I, IFAIL, J, N
*    .. Local Arrays ..
  complex       A(IA,NMAX)
  real          WKSPCE(NMAX)
*    .. External Subroutines ..
  EXTERNAL      F03ADF
*    .. Executable Statements ..
  WRITE (NOUT,*) 'F03ADF Example Program Results'
*    Skip heading in data file
  READ (NIN,*)
  READ (NIN,*) N
  WRITE (NOUT,*)
  IF (N.GE.0 .AND. N.LE.NMAX) THEN
    READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
    IFAIL = 0
*
    CALL F03ADF(A,IA,N,DETR,DETI,WKSPCE,IFAIL)
*
    WRITE (NOUT,99998) 'Value of determinant = (', DETR, ', ', DETI,
+      ' )'
  ELSE
    WRITE (NOUT,99999) 'N is out of range: N = ', N
  END IF
  STOP
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (1X,A,F9.4,A,F9.4,A)
  END

```

9.2 Program Data

F03ADF Example Program Data

```

3
(1.0, 0.0)  (1.0, 2.0)  (2.0,10.0)
(1.0, 1.0)  (0.0, 3.0)  (-5.0,14.0)
(1.0, 1.0)  (0.0, 5.0)  (-8.0,20.0)

```

9.3 Program Results

F03ADF Example Program Results

Value of determinant = (1.0000, 0.0000)
